

... TIC MARK**PURPOSE**

Specifies whether or not tic marks will appear on the frame lines of subsequent plots.

SYNTAX

<prefix>TIC MARK <ON or OFF>

where <prefix> is one of the following:

X1	refers to the bottom horizontal side;
X2	refers to the top horizontal side;
X	refers to both horizontal sides;
Y1	refers to the left vertical side;
Y2	refers to the right vertical side;
Y	refers to both vertical sides;
no prefix	refers to all 4 sides;

and ON specifies that the tic marks will appear while OFF specifies that they will not.

EXAMPLES

```
TIC MARK ON
TIC MARK OFF
TIC MARK
X2TIC MARK ON
Y2TIC MARK ON
```

NOTE 1

The appearance of the tic mark labels is independent of the appearance of the tic marks and is controlled with the TIC MARK LABEL command. However, no tics are drawn if the frame line is not drawn.

NOTE 2

A ...TIC MARK command with no arguments reverts the setting to default; thus X1TIC MARK with no arguments reverts the bottom horizontal tics to on. A ...TIC MARK command with no prefix refers to all 4 sides; thus TIC MARK OFF suppresses tic marks for all 4 frame lines. TIC MARK with no prefix and no arguments reverts the tic settings on all 4 sides to default.

DEFAULT

The lower horizontal and the left vertical frame lines have tics while the upper horizontal and right vertical frame lines do not.

SYNONYMS

TICS, TIC, and TICS MARK are synonyms for TIC MARK, as in TICS ON, TIC OFF, and X2TICS OFF.

RELATED COMMANDS

TIC MARK COLOR	=	Sets the color for tic marks.
TIC MARK OFFSET	=	Sets the offset for the first and last tic marks.
TIC MARK POSITION	=	Sets location of tic marks (inside/outside through).
TIC MARK SIZE	=	Sets the size of tic marks.
TIC MARK THICKNESS	=	Sets the thickness of tic marks.
TIC MARK LABEL	=	Specifies whether tic mark labels are drawn.

APPLICATIONS

Presentation Graphics

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```

LET ID2 = DATA 1 1 1 1 1 2 2 2 2 2 2 2 2 2 2 2
LET LEAD = DATA ...
      164 426 59 98 312 263 607 497 213 54 160 262 547 325 419 94 70
LET POT = DATA ...
      106 175 61 79 94 121 424 328 107 218 140 179 246 231 245 339 99
.
TITLE AUTOMATIC
TITLE SIZE 5
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
XILABEL LEAD
YILABEL POTASSIUM
LINE BLANK ALL
CHARACTER CIRCLE
CHARACTER FILL ON
PLOT POT LEAD
.
TIC MARKS ON
PLOT POT LEAD
.
TIC MARK OFF
PLOT POT LEAD
.
X1TIC MARK ON
X2TIC MARK OFF
Y1TIC MARK ON
Y2TIC MARK ON
PLOT POT LEAD
END OF MULTIPLY
    
```

